Quadratic Limit-state Function

To illustrate aspects of the second-order reliability method, SORM, we consider the following limit-state function:

```
g = 1818.0 + 15.34 x1 - 208.25 x2 + 0.04 x1^2 + 6.25 x2^2 - x1 x2;
```

The two random variables are uncorrelated and normally distributed with the following second-moment information:

```
μ1 = 50;
μ2 = 20;
σ1 = 5;
σ2 = 0.4;
```

Transformation into the standard normal space is simple without correlation:

```
x1fromy1 = μ1 + σ1 y1;
x2fromy2 = μ2 + σ2 y2;
```

That yields the following limit-state function in the standard normal space:

```
G = g /. {x1 -> x1fromy1, x2 → x2fromy2} // Expand
```

which yields: $20. - 3.3\,y1 + 1.\,y1^2 - 3.3\,y2 - 2.\,y1\,y2 + 1.\,y2^2$

The joint probability density function in the standard normal space, i.e., the bivariate standard normal PDF, is here plotted together with the limit-state function, *G*:

$$\varphi = \frac{1}{2\,\pi}\,\mathrm{Exp}\left[-\frac{y1^2 + y2^2}{2}\right];$$



The same two functions, φ and G, are here visualized in a contour plot, with a thick line to identify G=0; the following design point coordinates, determined by FORM, are also identified by a solid circle:

```
yStar = {3, 3};
```



The gradient vector in the standard normal space is:

```
∇G = {D[G, y1], D[G, y2]};
MatrixForm[∇G]
```

which yields: $\begin{pmatrix} -3.3 + 2.\ y1 - 2.\ y2 \\ -3.3 - 2.\ y1 + 2.\ y2 \end{pmatrix}$

That means the gradient vector at the design point is:

```
∇Gstar = ∇G /. {y1 → yStar[[1]], y2 → yStar[[2]]};
MatrixForm[∇Gstar]
```

which yields: $\begin{pmatrix} -3.3 \\ -3.3 \end{pmatrix}$

As a result the α-vector is:

```
α = - ∇Gstar / Norm[∇Gstar];
MatrixForm[α]
```

which yields: $\begin{pmatrix} 0.707107 \\ 0.707107 \end{pmatrix}$

On the basis of that α-vector a viable rotation matrix is:

```
P = {{-α[[1]], α[[1]]}, α};
MatrixForm[P]
```

which yields: $\begin{pmatrix} -0.707107 & 0.707107 \\ 0.707107 & 0.707107 \end{pmatrix}$

For the purpose of doing SORM analysis the Hessian, i.e., the second-order derivatives of the limit-state function is:

```
H = {D[∇G, y1], D[∇G, y2]};
MatrixForm[H]
```

which yields: $\begin{pmatrix} 2. & -2. \\ -2. & 2. \end{pmatrix}$

That means the **A**-matrix from the SORM theory is containing the zeros mentioned there:

```
A = P.H.P^T / Norm[∇Gstar];
Chop[MatrixForm[A]]
```

which yields: $\begin{pmatrix} 0.857099 & 0 \\ 0 & 0 \end{pmatrix}$

Because this problem only has two random variables there is no need for an eigenvalue analysis; there is only one curvature, κ_1, which is found in the (1,1) position of the **A**-matrix. The fact that κ_1 is positive implies that the limit-state surface curves outwards from the design point. This is seen in earlier plots. Now to the final results:

The reliability index from FORM is:

```
βFORM = Norm[yStar] // N
```

which yields: 4.24264

The associated failure probability from FORM is:

```
pfFORM = CDF[NormalDistribution[0, 1], -βFORM];
ScientificForm[pfFORM]
```

which yields: 1.10452×10^{-5}

The asymptotic SORM correction naturally yields a smaller result:

$$\text{pfSORM} = \text{pfFORM}\,\frac{1}{\sqrt{1 + \frac{\text{PDF[NormalDistribution[0,1],}\beta\text{FORM]}}{\text{pfFORM}}\,\text{A[[1, 1]]}}}$$

which yields: 5.03073×10^{-6}

That result is associated with the following generalized reliability index, not very different from the FORM index above:

```
-InverseCDF[NormalDistribution[0, 1], pfSORM]
```

which yields: 4.41585